// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/feeds/SwarmPriceFeed.sol



//SPDX-License-Identifier: GPL-3.0-only
pragma solidity ^0.8.17;

import "@openzeppelin/contracts/access/Ownable.sol";

contract SwarmPriceFeed is Ownable {
    string public description;

    uint256 private _latestOffchainTimestamp;

    uint256 public latestTimestamp;
    uint256 public latestAnswer;

    uint8 public decimals;

    event DescriptionUpdated(string oldDescription, string newDescription);
    event LatestAnswerUpdated(uint256 latestAnswer, uint256 latestTimestamp);
    event DecimalsUpdated(uint8 oldDecimals, uint8 newDecimals);

    constructor(
        string memory _initialDescription,
        uint256 _initialLatestAnswer,
        uint256 _initialLatestOffchainTimestamp,
        uint8 _initialDecimals
    ) {
        setDescription(_initialDescription);
        setLatestAnswer(_initialLatestAnswer, _initialLatestOffchainTimestamp);
        setDecimals(_initialDecimals);
    }

    function setDescription(string memory _description) public onlyOwner {
        emit DescriptionUpdated(description, _description);

        description = _description;
    }

    function setLatestAnswer(uint256 _answer, uint256 _offChaintimestamp) public onlyOwner {
        require(_offChaintimestamp > _latestOffchainTimestamp, "OFF_CHAIN_TIMESTAMP_TOO_OLD");
        require(_offChaintimestamp <= block.timestamp + 60, "OFF_CHAIN_TIMESTAMP_TOO_FAR_IN_FUTURE");

        emit LatestAnswerUpdated(_answer, block.timestamp);

        _latestOffchainTimestamp = _offChaintimestamp;

        latestAnswer = _answer;
        latestTimestamp = block.timestamp;
    }

    function setDecimals(uint8 _decimals) public onlyOwner {
        emit DecimalsUpdated(decimals, _decimals);

        decimals = _decimals;
    }

    function latestRoundData() external view returns (uint256, uint256, uint256, uint256, uint256) {
        return (0, latestAnswer, latestTimestamp, latestTimestamp, 0);
    }
}

// File: contracts/feeds/SwarmPriceFeedFactory.sol



//SPDX-License-Identifier: GPL-3.0-only
pragma solidity ^0.8.17;

import "@openzeppelin/contracts/access/Ownable.sol";

import "./SwarmPriceFeed.sol";

contract SwarmPriceFeedFactory is Ownable {
    event PriceFeedDeployed(address priceFeedAddress);

    function deployPriceFeed(
        string memory _description,
        uint256 _initialPrice,
        uint8 _decimals
    ) external onlyOwner returns (address) {
        SwarmPriceFeed swarmPriceFeed = new SwarmPriceFeed(_description, _initialPrice, block.timestamp, _decimals);

        swarmPriceFeed.transferOwnership(_msgSender());

        emit PriceFeedDeployed(address(swarmPriceFeed));

        return address(swarmPriceFeed);
    }
}